Funciones Función

Y

Funciones $\mathbb{R}^n \to \mathbb{R}^{m \times q}$

Prof.: Dr. Alejandro S. M. Santa Cruz
J.T.P.: Dr. Juan Ignacio Manassaldi
Aux. 1ra: Ing. Amalia Rueda

Funciones Función

Las ***Funciones "función"*** son funciones (**function**) que operan sobre otras funciones que son pasadas como argumentos de entrada a las primeras.

```
function y=ffun(f, a, b, ...)
//operaciones
endfunction
```

A diferencia de una function tradicional, aquí utilizamos como argumento otra function

Funciones Función

Un ejemplo clásico es la función **fplot**, que realiza la grafica de funciones.

Nombre de la Función que deseamos graficar

Limites de la grafica [xmin xmax]

Un ejemplo clásico es la función **fplot**, que realiza la grafica de funciones.

```
function fplot(f, lim)
x=linspace(lim(1),lim(2),1000);
plot(x,f(x))
endfunction
```

Ejemplo de aplicación:

```
function fplot(f, lim)
x=linspace(lim(1),lim(2),1000);
plot(x,f(x))
endfunction

function y=cuadratica(x)
y=x.^2;
endfunction

fplot(cuadratica,[-4,4])
```



- Para un correcto manejo de los algoritmos numéricos se deben definir diferentes funciones respetando dimensiones.
- Se analizaran funciones del tipo $\mathbb{R}^{n} \rightarrow \mathbb{R}^{m\times q}$
- Ejemplos:

$$f(x,y) = 2x^3 + 5y^2$$

$$f(x,y) = \begin{bmatrix} x^2 + y \\ -x + 3y^3 \end{bmatrix}$$

$$f(x,y) = \begin{bmatrix} 2x & x + y \\ 1 & x^2 + 3y \end{bmatrix}$$

- Resulta conveniente manipular las entradas como vectores y salidas de las funciones como matrices.
- Ejemplo:

$$f(x,y) = 2x^3 + 5y^2 \quad \rightarrow f(\underline{x}) = 2x_1{}^3 + 5x_2{}^2 \quad \underline{x} = \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$$

```
function f=fun(x)
f=2*x(1)^3+5*x(2)^2;
endfunction
```

```
--> fun([1,1])
 ans  =

   7.
```

- Ejemplo:

$$f(x,y)=\begin{bmatrix} x^2+y \\ -x+3y^3 \end{bmatrix} \quad \to f(\underline{x})=\begin{bmatrix} x_1^2+x_2 \\ -x_1+3x_2^3 \end{bmatrix} \quad \underline{x}=\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$$

```
function f=fun(x)
f(1,1)=x(1)^2+x(2);
f(2,1)=-x(1)+3*x(2)^3;
endfunction
```

```
function f=fun(x)
f=[x(1)^2+x(2)
-x(1)+3*x(2)^3];
endfunction
```

```
--> fun([1,1])
 ans  =

   2.
   2.
```

Funciones $\mathbb{R}^n \rightarrow \mathbb{R}^{m\times q}$

- Ejemplo:

$$f(x,y)=\begin{bmatrix} 2x & x+y \\ 1 & x^2+3y \end{bmatrix} \rightarrow f(\underline{x})=\begin{bmatrix} 2x_1 & x_1+x_2 \\ 1 & x_1{}^2+3x_2 \end{bmatrix} \qquad \underline{x}=\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$$

```
function f=fun(x)
f(1,1)=2*x(1);
f(1,2)=x(1)+x(2);
f(2,1)=1;
f(2,2)=x(1)^2+3*x(2)
endfunction
```

```
function f=fun(x)
f=[2*x(1)  x(1)+x(2)
     1      x(1)^2+3*x(2)];
endfunction
```

```
--> fun([1,1])
 ans  =

   2.   2.
   1.   4.
```

- Ejercicio: Crear una function que realice la grafica 3D de una función con los parámetros de color precargados.

```
function myplot3d(f, limx, limy)
x=linspace(limx(1),limx(2),500);
y=linspace(limy(1),limy(2),500);
for i=1:length(x)
        for j=1:length(y)
                z(i,j) = f([x(i);y(j)]);
        end
end
plot3d(x,y,z);
e = gce();
e.color_mode = -1
e.color_flag = 1;
g = gcf()
g.color_map= jetcolormap(32)
endfunction
```

- Ejercicio
  - Implementar en Scilab la función de Himmelblau.

$$f\left(\underline{x}\right)=\left(x_1^{\ 2}+x_2-11\right)^2+\left(x_1+x_2^{\ 2}-7\right)^2$$

  - Graficar en 3D para x_1=[-4,4] y x_2=[-4,4]
  - Graficar las siguiente curvas de nivel: 2, 5, 10, 20, 40, 60, 80, 100, 120, 140, 160, 170, 180

Funciones $\mathbb{R}^{n} \rightarrow \mathbb{R}^{m\times q}$

- Ejercicio: Implementar en Scilab la siguiente función, su vector gradiente y su matriz hessiana.

$$f\left(x,y\right)=3x^{2}y+5xe^{y} \rightarrow f\left(\underline{x}\right)=3{x_1}^{2}x_2+5x_1e^{x_2}$$

$$\nabla f=\begin{bmatrix} 6x_1x_2+5e^{x_2} \\ 3{x_1}^{2}+5x_1e^{x_2} \end{bmatrix}$$

$$H_f=\begin{bmatrix} 6x_2 & 6x_1+5e^{x_2} \\ 6x_1+5e^{x_2} & 5x_1e^{x_2} \end{bmatrix}$$